Offsetting Account Description in FBL3N and FAGLL03 – GL Line Item Display Reports

Applies to:

Organizations using SAP which need an additional field to be displayed in FBL3N & FAGLL03 reports.

Below configuration is based on SAP ECC 6.0. The configuration can also be held as reference for R/3 470 release and 46C release.

For more information, visit the Enterprise Resource Planning homepage.

Summary

Standard BTE functionality provides Offset Account Number & Offset Account Type. Additional fields can be pulled into the report with little modification, for a better reconciliation when displaying GL Line items.

Author: Imran Mohammad Khan. K

Company: YASH Technologies

Created on: 31 March 2011

Author Bio

Imran Khan is currently working as an FICO Consultant with YASH Technologies; Hyderabad.

Table of Contents

Overview

FBL3N and FAGLL03 for displaying GL Line items are the most frequently used reports. Most of the customers have a requirement of having an additional field in the report apart from the special fields in the line layout variant. This article explains the procedure for adding an additional field in the change layout of these reports. Users can create a layout and run the report using the layout.

Display GL Line Items

T. Code: FBL3N – GL Line Item Display



Output of FBL3N



BTE 1650 (Standard Functionality)

SAP provides **Offset Account Number (GKONT) & Offset Account Type (GKART)** fields to be displayed in the output of the report. This can be achieved using the BTE functionality in Financial Accounting.

Use T. Code: FIBF

Or

Menu Path

IMG → Financial Accounting (New) → Financial Accounting Global Settings (New) → Tools → Customer Enhancements → Business Transaction Events



The activation check box when checked lets the fields to appear in the layout.



Once the product of a customer is activated you need to create a P/S Module of a customer as shown in the above screen shot using **T. Code: FIBF**



Create new entries in the P/S module of a customer for BTE Event 00001650 and assign the product activated earlier, using **T. Code: FIBF.** Also assign the respective function module (LINE_ITEMS_GET_GKONT) and Save.

The following logic is used to populate the values in the required fields of the report.

```
FUNCTION LINE_ITEMS_GET_GKONT.
*"----------------------------------------------------------------------
*"*"Lokale Schnittstelle:
*"       IMPORTING
*"             VALUE(I_POSTAB) LIKE  RFPOS STRUCTURE  RFPOS
*"       EXPORTING
*"             VALUE(E_POSTAB) LIKE  RFPOS STRUCTURE  RFPOS
*"----------------------------------------------------------------------
  E_POSTAB = I_POSTAB.
```

```
    CALL FUNCTION 'GET_GKONT'
         EXPORTING
              BELNR           = I_POSTAB-BELNR
              BUKRS           = I_POSTAB-BUKRS
              BUZEI           = I_POSTAB-BUZEI
              GJAHR           = I_POSTAB-GJAHR
              GKNKZ           = '3'
         IMPORTING
              GKONT           = E_POSTAB-GKONT
              KOART           = E_POSTAB-GKART
         EXCEPTIONS
              BELNR_NOT_FOUND = 1
              BUZEI_NOT_FOUND = 2
              GKNKZ_NOT_FOUND = 3
              OTHERS          = 4.
    IF SY-SUBRC <> 0.
  * MESSAGE ID SY-MSGID TYPE SY-MSGTY NUMBER SY-MSGNO
  *         WITH SY-MSGV1 SY-MSGV2 SY-MSGV3 SY-MSGV4.
    ENDIF.

  ENDFUNCTION.
```

Once the BTE is activated, the standard fields GKONT & GKART appear in the Change Layout of the report.



Execute GL Line Item Display report using FBL3N to pull the Offset Account Number (GKONT) and Offset Account Type (GKART) from the change layout.



BTE 1650 with Additional Field

To add an additional field in the report, following steps need to be done in ABAP.

In our case we added **Offsetting Account Text** in the report layout.

Append the RFPOS & RFPOSX structures using T. Code: SE11 as shown in the following screen shots below.







Once the **RFPOS & RFPOSX** structures are appended, you need to ensure that both the structures are **ACTIVE.**

One Important step to be executed;

Using T. Code: SE38 to regenerate the program RFPOSEXTEND

After the program regeneration create another product of a customer to include the additional field.





The product of the customer with additional field logic should remain activated and the one for the standard fields can optionally be deactivated.



Create new entries in the P/S module of a customer for **BTE Event 00001650** and assign the product (OFFACTXT) created and activated earlier, using **T. Code: FIBF.**



Create a new function module **Z_LINE_ITEMS_GET_GKONT** and assign it to **BTE 00001650** and save.

Z_LINE_ITEMS_GET_GKONT

The following logic below is used to populate the values in the required fields of the report.

```
FUNCTION Z_LINE_ITEMS_GET_GKONT.
*"----------------------------------------------------------------------
*"*"Local Interface:
*"  IMPORTING
*"     VALUE(I_POSTAB) LIKE  RFPOS STRUCTURE  RFPOS
*"  EXPORTING
*"     VALUE(E_POSTAB) LIKE  RFPOS STRUCTURE  RFPOS
*"----------------------------------------------------------------------

  E_POSTAB = I_POSTAB.

  CALL FUNCTION 'GET_GKONT'
    EXPORTING
      BELNR           = I_POSTAB-BELNR
      BUKRS           = I_POSTAB-BUKRS
      BUZEI           = I_POSTAB-BUZEI
      GJAHR           = I_POSTAB-GJAHR
      GKNKZ           = '3'
    IMPORTING
      GKONT           = E_POSTAB-GKONT
      KOART           = E_POSTAB-GKART
    EXCEPTIONS
      BELNR_NOT_FOUND = 1
      BUZEI_NOT_FOUND = 2
      GKNKZ_NOT_FOUND = 3
      OTHERS          = 4.
  IF SY-SUBRC <> 0.
* MESSAGE ID SY-MSGID TYPE SY-MSGTY NUMBER SY-MSGNO
*         WITH SY-MSGV1 SY-MSGV2 SY-MSGV3 SY-MSGV4.
  Else.

    tables:
    skat,kna1,lfa1.
    Clear e_postab-zzgkont_ltxt.

* Get Offset Account Description Based on Account Type
    if sy-tcode = 'FBL3N'.
* In case of G/L Account
      if e_postab-gkart = 'S'.
        select single *
          from skat
          where spras = sy-langu and
                saknr = e_postab-gkont.
        if sy-subrc = 0.
          e_postab-zzgkont_ltxt = skat-TXT50.
        endif.
* In case of Customer
      elseif e_postab-gkart = 'D'.
        select single *
          from kna1
          where kunnr = e_postab-gkont.
        if sy-subrc = 0.
```

```
            e_postab-zzgkont_ltxt = kna1-name1.
          endif.
  * In case vendor
        elseif e_postab-gkart = 'K'.
          select single *
            from lfa1
            where lifnr = e_postab-gkont.
          if sy-subrc = 0.
            e_postab-zzgkont_ltxt = lfa1-name1.
          endif.
        endif.
      endif.
    ENDIF.

ENDFUNCTION.
```

The report is executed; showing the standard Offset Account Number, Offset Account Type and the additional field **Offsetting Account Description** that was added.



St	Assignment	DocumentNo	Type	Doc. Date	PK	Amount in local cur.	LCurr	Offst.acct	O	Name of offsetting account
	20091231	1800000000	DR	11/01/2009	01	2,750.00	USD	999101	S	De[redacted]a/c
	20091231	1800000001	DR	12/01/2009	01	2,750.00	USD	999101	S	D[redacted]/c
	20091231	1800000002	DR	12/01/2009	01	500.00	USD	999101	S	D[redacted]/c
	20091231	1800000003	DR	11/01/2009	01	500.00	USD	999101	S	D[redacted]/c
	20091231	1800000004	DR	11/01/2009	01	29,946.75	USD	999101	S	D[redacted]/c
	20091231	1800000005	DR	10/01/2009	01	9,780.00	USD	999101	S	D[redacted]/c
	20091231	1800000006	DR	07/01/2009	01	27,315.00	USD	999101	S	D[redacted]/c
	20091231	1800000007	DR	11/01/2009	01	6,454.00	USD	999101	S	D[redacted]/c
	20091231	1800000008	DR	12/01/2009	01	6,454.00	USD	999101	S	D[redacted]/c
	20091231	1800000009	DR	10/01/2009	01	6,454.00	USD	999101	S	D[redacted]/c
	20091231	1800000010	DR	11/01/2009	01	1,625.00	USD	999101	S	D[redacted]/c
	20091231	1800000011	DR	12/01/2009	01	1,625.00	USD	999101	S	D[redacted]/c
	20091231	1800000012	DR	11/01/2009	01	101,643.00	USD	999101	S	D[redacted]/c
	20091231	1800000013	DR	12/15/2009	01	24,444.00	USD	999101	S	D[redacted]/c
	20091231	1800000014	DR	12/31/2009	01	953.82	USD	999101	S	D[redacted]/c

BAdi 'Z_FAGL_ITEMS_CH_DATA'

In SAP ECC 6.0 you can use FAGLL03 in addition to FBL3N for GL account line item display. For displaying offset account info in FAGLL03 report, you must create and activate an implementation for the 'FAGL_ITEMS_CH_DATA' BAdI.

Following steps need to be carried out for creating and activating the BAdi implementation.

Call transaction SE19.



Choose Classic BAdi and enter the BAdi Name as 'FAGL_ITEMS_CH_DATA' and click on Create Impl. Button and create a new implementation.



As the Definition name select 'FAGL_ITEMS_CH_DATA'.

Enter an Implementation Name ‘Z_FAGL_ITEMS_CH_DATA’ as shown in the above screen shot.



On the 'Interface' tab select the 'CHANGE_ITEMS' method as shown in the screen shot above.

Logic for getting Offset Account Information in FAGLL03 report.

```
method IF_EX_FAGL_ITEMS_CH_DATA~CHANGE_ITEMS.
* Added Additional fields Offset A/C, Offset A/C type to Transaction
* FAGLL03
  DATA: wa_items TYPE faglposx.
  LOOP AT ct_items INTO wa_items.
    CALL FUNCTION 'GET_GKONT'
      EXPORTING
        belnr           = wa_items-belnr
        bukrs           = wa_items-bukrs
        buzei           = wa_items-buzei
        gjahr           = wa_items-gjahr
        gknkz           = '3'
      IMPORTING
        gkont           = wa_items-gkont
        koart           = wa_items-gkart
      EXCEPTIONS
        belnr_not_found = 1
        buzei_not_found = 2
        gknkz_not_found = 3
        OTHERS          = 4.
    IF sy-subrc = 0.

* Get Offset Account Description Based on Account Type
* In case of G/L Account
      Data: w_txt50 type txt50,
            w_name1 type name1.
      Clear: wa_items-zzgkont_ltxt,
             w_txt50,
```

```
                w_name1.
        if wa_items-gkart = 'S'.
          select single txt50
            from skat
            into w_txt50
            where spras = sy-langu and
                  saknr = wa_items-gkont.
          if sy-subrc = 0.
            wa_items-zzgkont_ltxt = w_TXT50.
          endif.
    * In case of Customer
        elseif wa_items-gkart = 'D'.
          select single name1
            from kna1
            into w_name1
            where kunnr = wa_items-gkont.
          if sy-subrc = 0.
            wa_items-zzgkont_ltxt = w_name1.
          endif.
    * In case vendor
        elseif wa_items-gkart = 'K'.
          select single name1
            from lfa1
            into w_name1
            where lifnr = wa_items-gkont.
          if sy-subrc = 0.
            wa_items-zzgkont_ltxt = w_name1.
          endif.
        endif.
        MODIFY ct_items FROM wa_items.
      ENDIF.
    ENDLOOP.
  endmethod.
```

The report is executed; showing the standard Offset Account Number, Offset Account Type and the additional field **Offsetting Account Description** that was added.



Related Content

Line item: Display of offsetting a/c information

Display of offsetting account information

For more information, visit the Enterprise Resource Planning homepage.

Disclaimer and Liability Notice